Here we investigate a RLC-circuit which is composed of two resistances, one coil, and one condenser. See J.-F. Pommaret, *Partial Differential Control Theory*, Kluwer, 2001, p. 576.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Weyl algebra $Alg = A_1$, where D is the differential operator w.r.t. time t. Note that we have to declare all constants that occur in the system matrix by using the option *comm* in the definition of the Ore Algebra *Alg*.

```
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t], comm=[R1, R2, C, L]):
```

We enter the system matrix R:

```
> R := evalm([[D+1/(R1*C), 0, -1/(R1*C)], [0, D+R2/L, -1/L]]);
```

$$R := \begin{bmatrix} \mathrm{D} + \frac{1}{R1\,C} & 0 & -\frac{1}{R1\,C} \\ 0 & \mathrm{D} + \frac{R2}{L} & -\frac{1}{L} \end{bmatrix}$$

The formal adjoint *R_adj* of R is computed using *Involution*:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} -\mathrm{D} + \frac{1}{R1\,C} & 0 \\ 0 & -\mathrm{D} + \frac{R2}{L} \\ -\frac{1}{R1\,C} & -\frac{1}{L} \end{bmatrix}$$

To check controllability of the RLC-circuit, we compute the first extension module extˆ1 with values in *Alg* of the *Alg*-module which is associated with *R_adj*:

```
> Ext := Exti(R_adj, Alg, 1);
```

$$Ext := \left[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} \mathrm{D}\,R1\,C + 1 & 0 & -1 \\ 0 & \mathrm{D}\,L + R2 & -1 \end{bmatrix}, \begin{bmatrix} \mathrm{D}\,L + R2 \\ \mathrm{D}\,R1\,C + 1 \\ \mathrm{D}\,L + L\,\mathrm{D}^2\,C\,R1 + R2\,C\,R1\,\mathrm{D} + R2 \end{bmatrix} \right]$$

Since $Ext[1]$ is the identity matrix, we see that the torsion submodule $\mathrm{t}(M)$ of the *Alg*-module M which is associated with the system is trivial *in the generic case*, i.e., the RLC-circuit is *generically* controllable (and, equivalently, parametrizable). This means that we have controllability for "almost all" choices of values for the constants $R1$, $R2$, C, L but there may be some configurations of the constants in which the RLC-circuit is not controllable. We shall actually find the only relation that leads to an uncontrollable system if the constants satisfy this relation.

Moreover, $Ext[3]$ is a parametrization of the system, namely we have $(x1 : x2 : u)^T = Ext[3]\,\xi$. We check whether this parametrization admits a left-inverse. In the affirmative case, we obtain a *flat output* of the system:

```
> S := LeftInverse(Ext[3], Alg);
```

$$S := \begin{bmatrix} -\frac{R1\,C}{L - R1\,C\,R2} & \frac{L}{L - R1\,C\,R2} & 0 \end{bmatrix}$$

Hence, whenever $R1\,C\,R2 - L \neq 0$, we find a flat output $\xi = S\ (x1 : x2 : u)^T$ of the system satisfies $(x1 : x2 : u)^T = Ext[3]\,\xi$. Then, the RLC-circuit is generically a flat system.

Let us point out that knowing the result of *Exti* above, it was already clear that the system is generically flat, because *Alg* is a principal ideal domain and torsion-free modules over principal ideal domains are free (and free modules correspond to flat systems).

The parametrization given in *Ext*[3], can be obtained directly by using *Parametrization*:

```
> Parametrization(R, Alg);
```

$$\begin{bmatrix} R2\,\xi_1(t) + L\,(\frac{d}{dt}\,\xi_1(t)) \\ \xi_1(t) + R1\,C\,(\frac{d}{dt}\,\xi_1(t)) \\ R2\,\xi_1(t) + L\,(\frac{d}{dt}\,\xi_1(t)) + (\frac{d}{dt}\,\xi_1(t))\,R1\,C\,R2 + L\,C\,R1\,(\frac{d^2}{dt^2}\,\xi_1(t)) \end{bmatrix}$$

In fact, one directly gets the last result by applying *Ext*[3] to the free function ξ_1:

```
> ApplyMatrix(Ext[3], [xi[1](t)], Alg);
```

$$\begin{bmatrix} R2\,\xi_1(t) + L\,(\frac{d}{dt}\,\xi_1(t)) \\ \xi_1(t) + R1\,C\,(\frac{d}{dt}\,\xi_1(t)) \\ R2\,\xi_1(t) + (L + R1\,C\,R2)\,(\frac{d}{dt}\,\xi_1(t)) + L\,C\,R1\,(\frac{d^2}{dt^2}\,\xi_1(t)) \end{bmatrix}$$

```
> SyzygyModule(R, Alg);
```

$$\mathrm{INJ}(2)$$

Therefore, R has full row rank. So we know that M is projective if and only if R admits a right-inverse:

```
> T := RightInverse(R, Alg);
```

$$T := \begin{bmatrix} \frac{L\,C\,R1}{L - R1\,C\,R2} & -\frac{L^2}{L - R1\,C\,R2} \\ \frac{R1^2\,C^2}{L - R1\,C\,R2} & -\frac{L\,C\,R1}{L - R1\,C\,R2} \\ \frac{\mathrm{D}\,R1^2\,C^2\,L}{L - R1\,C\,R2} + \frac{R1^2\,C^2\,R2}{L - R1\,C\,R2} & -\frac{C\,R1\,\mathrm{D}\,L^2}{L - R1\,C\,R2} - \frac{L^2}{L - R1\,C\,R2} \end{bmatrix}$$

```
> simplify(Mult(R, T, Alg));
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}$$

This right-inverse is well defined if and only if $R1\,C\,R2 - L \neq 0$. Hence, we recover the same condition on the constants as above. If $R1\,C\,R2 - L \neq 0$, then M is projective. Since the system is time-invariant, M is actually defined over the commutative polynomial ring with indeterminated D and t. By the theorem of Quillen and Suslin, we conclude that M is a free module, hence we see again that the system is flat.

Let us compute the Brunovský canonical form of the system in the case where $R1\,C\,R2 - L \neq 0$.

```
> B := Brunovsky(R, Alg);
```

$$B := \begin{bmatrix} -\frac{R1\,C}{L - R1\,C\,R2} & \frac{L}{L - R1\,C\,R2} & 0 \\ \frac{1}{L - R1\,C\,R2} & -\frac{R2}{L - R1\,C\,R2} & 0 \\ -\frac{1}{(L - R1\,C\,R2)\,C\,R1} & \frac{R2^2}{(L - R1\,C\,R2)\,L} & \frac{1}{L\,C\,R1} \end{bmatrix}$$

In other words, we have the following transformation between the system variables *x1*, *x2* and u and the Brunovský variables $z[i]$ and v:

```
> evalm([seq([z[i](t)],i=1..2),[v(t)]])=ApplyMatrix(B, [x1(t),x2(t),u(t)], Alg);
```

$$\begin{bmatrix} z_1(t) \\ z_2(t) \\ \mathrm{v}(t) \end{bmatrix} = \begin{bmatrix} -\frac{R1\,C\,\mathrm{x1}(t)}{L - R1\,C\,R2} + \frac{L\,\mathrm{x2}(t)}{L - R1\,C\,R2} \\ \frac{\mathrm{x1}(t)}{L - R1\,C\,R2} - \frac{R2\,\mathrm{x2}(t)}{L - R1\,C\,R2} \\ -\frac{\mathrm{x1}(t)}{(L - R1\,C\,R2)\,C\,R1} + \frac{R2^2\,\mathrm{x2}(t)}{(L - R1\,C\,R2)\,L} + \frac{\mathrm{u}(t)}{L\,C\,R1} \end{bmatrix}$$

Let us check that the new variables $z[\,i]$ and v satisfy a Brunovský canonical form:

```
> F := Elimination(linalg[stackmatrix](B, R), [x1,x2,u],
> [seq(z[i],i=1..2),v,0,0], Alg):
> ApplyMatrix(F[1], [x1(t),x2(t),u(t)], Alg)=
> ApplyMatrix(F[2], [seq(z[i](t),i=1..2),v(t)], Alg);
```

$$\begin{bmatrix} 0 \\ 0 \\ \mathrm{u}(t) \\ \mathrm{x2}(t) \\ \mathrm{x1}(t) \end{bmatrix} = \begin{bmatrix} -(\frac{d}{dt}\, z_2(t)) + \mathrm{v}(t) \\ -(\frac{d}{dt}\, z_1(t)) + z_2(t) \\ R2\, z_1(t) + (R1\,C\,R2 + L)\, z_2(t) + L\,C\,R1\,\mathrm{v}(t) \\ z_1(t) + R1\,C\, z_2(t) \\ R2\, z_1(t) + L\, z_2(t) \end{bmatrix}$$

Now, we study the case where $R1\,C\,R2 = L$.

```
> R_mod := subs(L=R1*R2*C, evalm(R));
```

$$R\_mod := \begin{bmatrix} \mathrm{D} + \frac{1}{R1\,C} & 0 & -\frac{1}{R1\,C} \\ 0 & \mathrm{D} + \frac{1}{R1\,C} & -\frac{1}{R1\,C\,R2} \end{bmatrix}$$

```
> Ext_mod := Exti(Involution(R_mod, Alg), Alg, 1);
```

$$Ext\_mod := \left[\begin{bmatrix} \mathrm{D}\,R1\,C + 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & -R2 & 0 \\ 0 & R2\,R1\,C\,\mathrm{D} + R2 & -1 \end{bmatrix}, \begin{bmatrix} -R2 \\ -1 \\ -R2\,R1\,C\,\mathrm{D} - R2 \end{bmatrix} \right]$$

Hence, if $R1\,C\,R2 = L$, then the RLC-circuit is not controllable. The generating set of torsion elements (which correspond to autonomous elements of the system) can also be obtained using *TorsionElements*. The generating torsion elements are expressed in terms of the dependent variables *x1*, *x2*, u of the system:

```
> TorsionElements(R_mod, [x1(t),x2(t),u(t)], Alg);
```

$$[[\ \theta_1(t) + R1\,C\,(\tfrac{d}{dt}\,\theta_1(t)) = 0\ ],\ [\ \theta_1(t) = \mathrm{x1}(t) - R2\,\mathrm{x2}(t)\ ]]$$

We know that the torsion elements and the first integrals of motion are in one-to-one correspondence. Hence, let us compute a first integral of motion of the system.

```
> V := FirstIntegral(R_mod, [x1(t),x2(t),u(t)], Alg);
```

$$V := -\frac{\_C1\, e^{(\frac{t}{R1\,C})}\,(\mathrm{x1}(t) - R2\,\mathrm{x2}(t))}{R2}$$

Let us check that the derivative of V with respect to t is 0.

```
> Vdot := diff(V,t);
```

$$Vdot := -\frac{\_C1\, e^{\left(\frac{t}{R1\, C}\right)}\, (\mathrm{x1}(t) - R2\, \mathrm{x2}(t))}{R1\, C\, R2} - \frac{\_C1\, e^{\left(\frac{t}{R1\, C}\right)}\, ((\frac{d}{dt}\, \mathrm{x1}(t)) - R2\, (\frac{d}{dt}\, \mathrm{x2}(t)))}{R2}$$

The equations of the system are given by:

```
> Sys_mod := ApplyMatrix(R_mod, [x1(t),x2(t),u(t)], Alg) = evalm([[0],[0]]);
```

$$Sys\_mod := \begin{bmatrix} \frac{\mathrm{x1}(t)}{R1\, C} + (\frac{d}{dt}\, \mathrm{x1}(t)) - \frac{\mathrm{u}(t)}{R1\, C} \\ \frac{\mathrm{x2}(t)}{R1\, C} + (\frac{d}{dt}\, \mathrm{x2}(t)) - \frac{\mathrm{u}(t)}{R1\, C\, R2} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

From these equations, let us extract the first derivatives of *x1* and *x2*:

```
> lhs1 := solve(lhs(Sys_mod)[1,1], diff(x1(t),t));
```

$$lhs1 := -\frac{\mathrm{x1}(t) - \mathrm{u}(t)}{R1\, C}$$

```
> lhs2 := solve(lhs(Sys_mod)[2,1], diff(x2(t),t));
```

$$lhs2 := -\frac{R2\, \mathrm{x2}(t) - \mathrm{u}(t)}{R1\, C\, R2}$$

```
> simplify(subs({diff(x1(t),t)=lhs1, diff(x2(t),t)=lhs2}, Vdot));
```

$$0$$

Therefore, if we take into account the equations of the systems, *Vdot* becomes 0, i.e., V is a first integral of motion.

Finally, let us compute the parametrization of the system defined by the matrix R by means of an arbitrary function ξ_1 and constants:

```
> P := Parametrization(R_mod, Alg);
```

$$P := \begin{bmatrix} \_C1\, e^{\left(-\frac{t}{R1\, C}\right)} - R2\, \xi_1(t) \\ -\xi_1(t) \\ -R2\, (\xi_1(t) + R1\, C\, (\frac{d}{dt}\, \xi_1(t))) \end{bmatrix}$$

We can check that P parametrizes some solutions of the system defined by R:

```
> simplify(ApplyMatrix(R_mod, P, Alg));
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

In fact, we can prove that P parametrizes all the C^{∞}solutions. For more details, we refer the reader to A. Quadrat, D. Robertz, "*On Monge problem for uncontrollable linear systems*", to appear.

In (Pommaret, 2001) more examples for electric circuits are presented which can be treated analogously by using *OreModules*.